

NAG Library Routine Document

F08QTF (ZTREXC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08QTF (ZTREXC) reorders the Schur factorization of a complex general matrix.

2 Specification

```
SUBROUTINE F08QTF (COMPQ, N, T, LDT, Q, LDQ, IFST, ILST, INFO)
INTEGER N, LDT, LDQ, IFST, ILST, INFO
COMPLEX (KIND=nag_wp) T(LDT,*), Q(LDQ,*)
CHARACTER(1) COMPQ
```

The routine may be called by its LAPACK name *ztrexc*.

3 Description

F08QTF (ZTREXC) reorders the Schur factorization of a complex general matrix $A = QTQ^H$, so that the diagonal element of T with row index IFST is moved to row ILST.

The reordered Schur form \tilde{T} is computed by a unitary similarity transformation: $\tilde{T} = Z^HTZ$. Optionally the updated matrix \tilde{Q} of Schur vectors is computed as $\tilde{Q} = QZ$, giving $A = \tilde{Q}\tilde{T}\tilde{Q}^H$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- | | | |
|----|---|---------------------|
| 1: | COMPQ – CHARACTER(1) | <i>Input</i> |
| | <i>On entry:</i> indicates whether the matrix Q of Schur vectors is to be updated. | |
| | COMPQ = 'V' | |
| | The matrix Q of Schur vectors is updated. | |
| | COMPQ = 'N' | |
| | No Schur vectors are updated. | |
| | <i>Constraint:</i> COMPQ = 'V' or 'N'. | |
| 2: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the order of the matrix T . | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 3: | T(LDT,*) – COMPLEX (KIND=nag_wp) array | <i>Input/Output</i> |
| | Note: the second dimension of the array T must be at least $\max(1, N)$. | |
| | <i>On entry:</i> the n by n upper triangular matrix T , as returned by F08PSF (ZHSEQR). | |
| | <i>On exit:</i> T is overwritten by the updated matrix \tilde{T} . | |

4:	LDT – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array T as declared in the (sub)program from which F08QTF (ZTREXC) is called.		
<i>Constraint:</i> $LDT \geq \max(1, N)$.		
5:	Q(LDQ, *) – COMPLEX (KIND=nag_wp) array	<i>Input/Output</i>
Note: the second dimension of the array Q must be at least $\max(1, N)$ if $\text{COMPQ} = 'V'$ and at least 1 if $\text{COMPQ} = 'N'$.		
<i>On entry:</i> if $\text{COMPQ} = 'V'$, Q must contain the n by n unitary matrix Q of Schur vectors.		
<i>On exit:</i> if $\text{COMPQ} = 'V'$, Q contains the updated matrix of Schur vectors.		
If $\text{COMPQ} = 'N'$, Q is not referenced.		
6:	LDQ – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array Q as declared in the (sub)program from which F08QTF (ZTREXC) is called.		
<i>Constraints:</i>		
if $\text{COMPQ} = 'V'$, $LDQ \geq \max(1, N)$; if $\text{COMPQ} = 'N'$, $LDQ \geq 1$.		
7:	IFST – INTEGER	<i>Input</i>
8:	ILST – INTEGER	<i>Input</i>
<i>On entry:</i> IFST and ILST must specify the reordering of the diagonal elements of T. The element with row index IFST is moved to row ILST by a sequence of exchanges between adjacent elements.		
<i>Constraint:</i> $1 \leq \text{IFST} \leq N$ and $1 \leq \text{ILST} \leq N$.		
9:	INFO – INTEGER	<i>Output</i>
<i>On exit:</i> INFO = 0 unless the routine detects an error (see Section 6).		

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix \tilde{T} is exactly similar to a matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

The values of the eigenvalues are never changed by the reordering.

8 Parallelism and Performance

F08QTF (ZTREXC) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $20nr$ if $\text{COMPQ} = \text{'N'}$, and $40nr$ if $\text{COMPQ} = \text{'V}'$, where $r = |\text{IFST} - \text{ILST}|$.

The real analogue of this routine is F08QFF (DTREXC).

10 Example

This example reorders the Schur factorization of the matrix T so that element t_{11} is moved to t_{44} , where

$$T = \begin{pmatrix} -6.00 - 7.00i & 0.36 - 0.36i & -0.19 + 0.48i & 0.88 - 0.25i \\ 0.00 + 0.00i & -5.00 + 2.00i & -0.03 - 0.72i & -0.23 + 0.13i \\ 0.00 + 0.00i & 0.00 + 0.00i & 8.00 - 1.00i & 0.94 + 0.53i \\ 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i & 3.00 - 4.00i \end{pmatrix}.$$

10.1 Program Text

```
Program f08qtf
!
!     F08QTF Example Program Text
!
!     Mark 26 Release. NAG Copyright 2016.
!
!     .. Use Statements ..
Use nag_library, Only: nag_wp, x04dbf, ztrexc
!
!     .. Implicit None Statement ..
Implicit None
!
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!
!     .. Local Scalars ..
Integer :: i, ifail, ifst, ilst, info, ldq,      &
            ldt, n
!
!     .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: q(:,:), t(:,:)
Character (1) :: clabs(1), rlabs(1)
!
!     .. Executable Statements ..
Write (nout,*) 'F08QTF Example Program Results'
Write (nout,*)
Flush (nout)
!
!     Skip heading in data file
Read (nin,*)
Read (nin,*) n
ldq = 1
ldt = n
Allocate (q(ldq,1),t(ldt,n))
!
!     Read T from data file
!
Read (nin,*)(t(i,1:n),i=1,n)
!
!     Read the row indices
!
Read (nin,*) ifst, ilst
!
!     Reorder the Schur factor T
!
!     The NAG name equivalent of ztrexc is f08qtf
Call ztrexc('No update',n,t,ldt,q,ldq,ifst,ilst,info)
If (info/=0) Then
    Write (nout,99999) info, ilst
    Write (nout,*)
    Flush (nout)
```

```

End If

!      Print reordered Schur form

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04dbf('General',' ',n,n,t,ldt,'Bracketed','F7.4',
             'Reordered Schur form','Integer',rlabs,'Integer',clabs,80,0,ifail)      &

99999 Format (' Reordering could not be completed. INFO = ',I3,' ILST = ',I5)
End Program f08qtf

```

10.2 Program Data

```

F08QTF Example Program Data
 4                                         :Value of N
(-6.00,-7.00) ( 0.36,-0.36) (-0.19, 0.48) ( 0.88,-0.25)
( 0.00, 0.00) (-5.00, 2.00) (-0.03,-0.72) (-0.23, 0.13)
( 0.00, 0.00) ( 0.00, 0.00) ( 8.00,-1.00) ( 0.94, 0.53)
( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 3.00,-4.00) :End of matrix T
 1   4                                         :Values of IFST and ILST

```

10.3 Program Results

F08QTF Example Program Results

Reordered Schur form				
	1	2	3	4
1	(-5.0000, 2.0000)	(-0.1574, 0.7143)	(0.1781,-0.1913)	(0.3950, 0.3861)
2	(0.0000, 0.0000)	(8.0000,-1.0000)	(1.0742, 0.1447)	(0.2515,-0.3397)
3	(0.0000, 0.0000)	(0.0000, 0.0000)	(3.0000,-4.0000)	(0.2264, 0.8962)
4	(0.0000, 0.0000)	(0.0000, 0.0000)	(0.0000, 0.0000)	(-6.0000,-7.0000)
